

PDF mit SAS 8.2

Heinrich Stürzl

Dade Behring Marburg GmbH

Statistical Systems & Data Management

Heinrich_Stuerzl@DadeBehring.com

Zusammenfassung

PDF (Portable Document Format) ist das gängige Format zur Weitergabe von Dokumenten per Email und im Intranet bzw. Internet. Es werden vier verschiedene Möglichkeiten beschrieben und verglichen, wie mit der SAS Version 8.2 Listen- und Grafikoutput als PDF-Datei gespeichert werden kann. Vorgestellt werden zwei Möglichkeiten mittels **Acrobat Distiller** unter Windows (ohne und mit ODS), sowie die beiden generischen Methoden **ODS PS** und **ODS PDF**, welche mit SAS unabhängig vom jeweiligen Betriebssystem zur Verfügung stehen¹.

Keywords: PDF, SAS, Version 8, Acrobat Distiller, ODS PRINTER, Postscript, PS.

1 Einleitung

Bei den Methoden zur Erstellung von PDF-Dateien ist grundsätzlich zwischen ODS und non-ODS Methoden einerseits und den verschiedenen Betriebssystemen andererseits zu unterscheiden. Unter Windows kann die Erstellung von PDF-Dateien mit Hilfe der Software Acrobat der Firma Adobe erfolgen. Darin sind zwei verschiedene Möglichkeiten zur Erstellung enthalten.

1. „**Acrobat PDFWriter**“, ein Druckertreiber, der ein Dokument direkt aus der Anwendung konvertiert (zeigt geringere Qualität und Auflösung und wird hier nicht beschrieben).
2. „**Acrobat Distiller**“ konvertiert Postscript (PS) Dateien und erzielt bessere Ergebnisse.

¹Siehe auch Tipps & Tricks der 5. KSFE 2001 in Hohenheim.

Alternativ kann man **Ghostscript** verwenden, um Postscript Dateien in PDF umzuwandeln (wird hier nicht beschrieben):

<http://www.cs.wisc.edu/~ghost/>

Damit lässt sich SAS Output wie bei jeder anderen Windows Anwendung durch einfaches „Drucken“ als PDF-Datei speichern. Entscheidend dabei ist, dass der Output zuerst erstellt und in SAS angezeigt werden muss. Erst danach kann entschieden werden, ob er „gedruckt“ werden soll. Dieser Weg benötigt kein ODS und ist folglich auch nicht an Version 8 gebunden.

Eine andere non-ODS Möglichkeit ist der SAS/GRAPH Treiber **PDFC**, welcher betriebssystemunabhängig, jedoch auf Grafiken beschränkt ist. Damit lassen sich die Grafiken einer einzelnen Prozedur wie ein Grafikstreamfile als PDF-Datei speichern (wird hier nicht weiter beschrieben).

Mit Einführung des Output Delivery System (ODS) in Version 8 stehen mit **ODS PRINTER** grundsätzlich neue Möglichkeiten zur Verfügung.

ODS PRINTER erzeugt direkt eine Ausgabedatei (Postscript, PDF, ...) für den Output beliebiger Prozeduren. Dieses führt zur Gleichbehandlung von Listen- und Grafikoutput. Listenoutput, also Output von Nicht-Grafik-Prozeduren, der normalerweise im Output (Listen) Fenster angezeigt wird, wird dabei hochaufgelöst als formatierte Tabelle erstellt.

Das Layout der Datei wird an verschiedenen Stellen in der SAS Software festgelegt, wobei diese in folgender Reihenfolge abgearbeitet werden:

1. STYLE= Option von ODS PRINTER
wählt ein bestimmtes Template aus, worin Schriftarten, -größen, Farben etc. festgelegt sind (Standard ist STYLE=PRINTER)
2. GOPTIONS
unterstützt wird: FTITLE, FTEXT, GUNIT, HTITLE, HTEXT
3. Optionen in TITLE/FOOTNOTE Anweisung
unterstützt wird: FONT, HEIGHT, COLOR, UNDERLIN

Darüber hinaus beeinflussen auch noch einige Systemoptionen ODS PRINTER (z.B. CENTER, ORIENTATION=, PAPERSIZE=, LEFTMARGIN=), allerdings nur dann, wenn diese vor der ODS Anweisung definiert werden.

SAS/GRAPH Software Fonts wie SIMPLEX, SWISS etc. stehen auch für ODS PRINTER zur Verfügung, führen jedoch zu weniger guten Ergebnissen als die True Type Schriften (Hardware Fonts), die in der SAS Registry (Solutions ⇒ Accessories ⇒ Registry Editor unter \CORE\PRINTING\PSL\FONT FAMILIES eingetragen sind. Von diesen beherrschen allerdings bisher ausschließlich „**Helvetica**“ und „**Times**“ die deutschen Umlaute. True Type Schriften werden als solche in die Druckerdatei geschrieben, wodurch Text im Ausdruck schärfer

und die Datei kleiner wird. Allerdings sind sie noch problematisch bei gedrehten Texten (z.B. A=90), was in Version 9 besser werden soll. Als Ersatz können hier Software Fonts verwendet werden. Siehe SAS Note SN-005344.

Falls Grafiken zusammen mit Listenoutput in eine PDF-Datei geschrieben werden sollen, ist zu beachten, dass alle mit GOPTIONS definierten Grafikoptionen ausschließlich für die Grafiken gelten. Zum Beispiel

```
GOPTIONS FTEXT="Helvetica" HTEXT=12 PT CTEXT=RED;
TITLE "Überschrift";
```

Da diese Optionen bei TITLE und FOOTNOTE Anweisungen nur im Grafikoutput umgesetzt und im Listenoutput ignoriert werden, wird bei einem einfachen PROC PRINT die Warnung ausgegeben, dass keine Schriftart gefunden wurde. Stattdessen wird die Defaultschriftart verwendet, die anders aussieht und keine Umlaute darstellen kann.

```
WARNING: Font specification "UNKNOWN" failed.
        Using default font family instead.
```

Das Problem kann umgangen werden, indem in jeder TITLE und FOOTNOTE Anweisung Schriftart, -größe und -farbe explizit gesetzt wird.

```
TITLE F="Helvetica" H=12 PT C=RED "Überschrift";
```

Bei Grafiken ist zu beachten, dass die **Seitenorientierung** mit der Grafik Rotation (LANDSCAPE, PORTRAIT) übereinstimmt. Andernfalls können Titel und Fußnoten ohne Warnung vollständig fehlen. Das Problem soll in Version 9 behoben sein. Siehe SAS Note SN-005344.

```
GOPTIONS ROTATE=LANDSCAPE;
OPTIONS ORIENTATION=LANDSCAPE;
```

bzw.

```
GOPTIONS ROTATE=PORTRAIT;
OPTIONS ORIENTATION=PORTRAIT;
```

Da OPTIONS Anweisungen nach einer ODS PRINTER Anweisung generell wirkungslos sind, kann die Seitenorientierung innerhalb einer PDF-Datei nicht geändert werden.

Ab SAS Version 8.2 gibt es neue ODS Kanäle (destinations), die Synonyme für bisherige ODS PRINTER Optionen sind, diese jedoch langfristig ersetzen sollen.

- ODS PS = ODS PRINTER PS
- ODS PDF = ODS PRINTER PDF

Beide Wege nutzen den sog. „**Universal Printing mode**“ („SAS mode“), welcher unabhängig vom Betriebssystem arbeitet. Die so erstellten PDF-Dateien richten sich nach den Einstellungen der SAS Registry und verwenden deshalb standardmäßig das Letter Format. Durch Verwendung der Systemoption **PAPERSIZE=**„ISO A4“ kann jedoch auf A4 Format umgestellt werden (siehe auch SN-005913).

Im Gegensatz dazu gibt es unter Windows bei ODS PRINTER auch den „**PC host mode**“, bei dem die Windows Registry maßgeblich ist. Dies gilt dann, wenn in der PRINTER= Option ein in Windows installierter Drucker angegeben wird (kann auch „Acrobat Distiller“ sein). Die Seiteneinstellungen lassen sich dann im „Print Setup“ Dialog konfigurieren.

Quellen

Da das gesamte Umfeld von ODS PRINTER noch vergleichsweise neu und in Entwicklung begriffen ist, treten an verschiedenen Stellen noch Probleme auf. Eine gute Übersicht über den aktuellen Stand der Technik bietet das FAQ Dokument des Systementwicklers Brian Schellenberger, SAS, Cary
<http://www.sas.com/rnd/base/topics/odsprinter/faq.html>

Siehe auch:

- Changes & Enhancements for ODS by Example (through Version 8.2), von Sandy McNeill, SAS, Cary
- TS-659, Exporting SAS/GRAPH Output to PDF Files from Release 8.2 and higher
- New Features for ODS destinations in 8.2
<http://www.sas.com/rnd/base/topics/odsprinter/new82.html>

Testbedingungen

Zum Test wurde ein einfaches SAS Programm verwendet (siehe Seite 305), welches ein mehrseitiges Datenlisting mit PROC PRINT im Hochformat und zwei Balkendiagramme mit PROC GCHART per BY Anweisung im Querformat erzeugt. Die Titel- und Fußzeilen enthalten deutsche Umlaute. Die Fußzeile wird in der Grafik als vertikaler Text dargestellt. Zur Überprüfung der korrekten Darstellung der ersten und letzten Spalte werden Titel und Fußzeilen im Listing linksbündig ausgerichtet.

Verwendete Software

- Windows NT 4.0 SP5
- SAS Version 8.2 TS2M0 ohne Hotfixes
- Adobe Acrobat 4.0 (deutsch)
- Distillereinstellungen: MonitorOptimiert(Schriften einbetten), wobei nur SASMonospace und SASMonospaceBold eingebettet wird. Standardpapierformat auf A4 Seitengröße (21,0 cm x 29,7 cm) eingestellt.
- Druckereinstellung in SAS (File ⇒ Print Setup ...): Schriftart „SAS Monospace“ 7 pt, Hochformat und 5 mm Seitenränder oben, unten und rechts, sowie ein Lochungsrand links von 10 mm
- Grafikschriftarten: SAS/GRAPH Software Fonts SWISS, SIMPLEX bzw. True Type Fonts „Helvetica“ und „Times“

2 Distiller (Windows, ohne ODS)

Der Distiller wandelt die erstellten PS-Dateien in PDF-Dateien. Diese werden in einem zentralen Verzeichnis abgelegt, welches in den Distiller Eigenschaften in der Lasche „Anschlüsse“ unter Einstellungen ⇒ Drucker konfiguriert werden kann. Zum Beispiel: C:\Temp*.pdf

Vorgehensweise:

1. Im Print Setup Dialog „Acrobat Distiller“ als Drucker auswählen und gewünschte Schriftart, Schriftgröße und Ränder festlegen, um dadurch die richtige Einstellung für Linesize und Pagesize zu bekommen. Alternativ kann dies auch per Programm mit entsprechenden System Optionen geschehen.
2. Anschließend das Programm starten und den Listen- bzw. Grafikoutput erzeugen. Dieser kann anschließend in eine PDF-Datei „gedruckt“ werden (Postscriptdatei wird automatisch in PDF umgewandelt), wobei der Dateiname automatisch vorgegeben wird. Falls eine Datei dieses Namens bereits existiert, wird automatisch ein neuer Name durch Anhängen einer geklammerten Nummer vorgeschlagen, den man annehmen oder ändern kann. Der Dateiname selbst hängt allerdings von dem Fenster ab, aus dem 'gedruckt' wird. Die folgende Tabelle gibt eine Übersicht.

„Drucken“ aus	Standardname der Datei	Datei- größe [KB] ^a	Hinweise/Probleme in PDF-Datei
Listenfenster	<i>SAS-Output.pdf</i>	20	Falsche Seitenränder^b
Resultsfenster Listing	<i>SAS-Results.pdf</i>	20	Footnotes fehlen auf der letzten Seite!^c
Resultsfenster Grafik	<i>SAS-SAS Graph</i> <i>Generic Driver.pdf</i>	64	Grafik erscheint am Bild- schirm im Hochformat ^d , eine Datei pro Grafik
Graphfenster	<i>SAS-SAS Graph</i> <i>Generic Driver.pdf</i>	64	Grafik erscheint am Bild- schirm im Hochformat eine Datei pro Grafik

^aAlle Dateigrößen bezogen auf WinNT und das verwendete Beispiel (s. S. 305)

^bGenerelles Problem unter Windows (95/98/NT/2000) beim „Drucken“ aus dem Listenfenster (Output). Für die korrekte Einhaltung der in SAS spezifizierten Seitenränder muss die Systemoption **-NOENABLEDUPRINTOPTIONS** in der SASV8.CFG gesetzt oder Hotfix 82ba46 eingespielt sein. Siehe SAS Note SN-**005042**.

^cGenerelles Problem unter Windows (95/98/NT/2000) beim Drucken eines Texteintrags aus dem Results Fenster. Siehe SN-**004327**. Soll in V9 behoben sein.

^dQuerformatgrafiken (GOPTIONS ROTATE=LANDSCAPE) erscheinen in der PDF-Datei im Hochformat (siehe SN-**003008**). **V8 Lösung: Mindestens im ersten Titel (in weiteren Texten optional) eine True Type Schrift z.B. „Helvetica“ verwenden!**

3 ODS PRINTER PRINTER='Acrobat Distiller' (Windows)

Setzt voraus, dass der Acrobat Distiller unter Windows installiert ist. Erstellt eine PS-Datei im „PC host mode“, wobei die Seiteneinstellungen konfigurierbar sind.

Vorgehensweise:

1. PS-Datei in ODS PRINTER Anweisung festlegen. Optional kann über die Option STYLE= das Layout beeinflusst werden. Standardmäßig wird STYLE=PRINTER verwendet.

```
ODS PRINTER PRINTER='Acrobat Distiller' FILE='C:\test2.ps';
```
2. Der Listen- und Grafikoutput aller nachfolgend aufgerufenen Prozeduren wird in die angegebene PS-Datei geschrieben, bis diese explizit geschlossen wird. Wichtig ist es, Prozeduren mit RUN bzw. QUIT vor dem ODS CLOSE zu beenden, sonst fehlt der Output der letzten Prozedur in der PS-Datei.

```
ODS PRINTER CLOSE;
```
3. Nachträglich PS-Datei in PDF-Datei z.B. mit dem Distiller umwandeln.

Hinweise:

1. SAS/GRAPH Software Fonts wie SIMPLEX, SWISS etc. sind problemlos verwendbar (auch in gedrehten Texten), führen jedoch zu stark vergrößerten PDF-Dateien. Mit True Type Fonts (Hardware Fonts) können Probleme auftreten, v.a. bei gedrehten Texten (z.B. ANGLE=90). Siehe SAS Note SN-005344

Ergebnis:

- Alle Outputs kommen in eine PDF-Datei, deren Größe hängt auch von der verwendeten Style Option ab (im Beispiel 37 KB mit True Type Fonts, 546 KB mit Font SWISS)
- Listenoutput wird hochaufgelöst als Tabelle erstellt abhängig von der verwendeten Style Option.
- PDF-Datei wird nicht direkt erstellt (PS-Datei muss nachträglich umgewandelt werden)

4 ODS PS

Erstellt eine PS-Datei mit Hilfe des SAS Postscript Treibers im „Universal Printing mode“ (identisch mit ODS PRINTER PS).

Vorgehensweise:

1. PS-Datei in ODS PS Anweisung festlegen.
Optional kann über die Option STYLE= das Layout beeinflusst werden. Standardmäßig wird STYLE=PRINTER verwendet.
`ODS PS FILE='C:\test2b.ps';`
2. Der Listen- und Grafikoutput aller nachfolgend aufgerufenen Prozeduren wird in die angegebene PS-Datei geschrieben, bis diese explizit geschlossen wird. Wichtig ist es, Prozeduren mit RUN bzw. QUIT vor dem ODS CLOSE zu beenden, sonst fehlt der Output der letzten Prozedur in der PDF-Datei.
`ODS PS CLOSE;`
3. Nachträglich PS-Datei in PDF-Datei z.B. mit dem Distiller umwandeln.

Hinweise:

2. Standardmäßig wird bei ODS PS das Letter Format verwendet, was in der PDF-Datei zu großen Seitenrändern und damit zu einer Verkleinerung des bedruckbaren Bereichs führt. Mit Hilfe der Systemoption **PAPERSIZE="ISO A4"** (vor der ODS PS Anweisung gesetzt sein) kann die PS-Datei im A4 Format erstellt werden. Damit dies auch bei der Umwandlung in die PDF-Datei korrekt umgesetzt wird, muss im Acrobat Distiller die **A4 Seitengröße (21,0 cm x 29,7 cm) als Standard** eingestellt sein (Voreinstellungen ⇒ Einstellungen ⇒ Erweitert ⇒ Standardpapierformat).
3. **Gedrehte Texte** (z.B. ANGLE=90) mit Font „Helvetica“ nicht darstellbar (wird automatisch durch Font „Simulate“ ersetzt, Umlaute OK). Das Problem soll in Version 9 behoben sein. Siehe SAS Note SN-005344.

Ergebnis:

- Alle Outputs kommen in eine PDF-Datei, deren Größe hängt auch von der verwendeten Style Option ab (im Beispiel 65 KB)
- Listenoutput wird hochaufgelöst als Tabelle erstellt abhängig von der verwendeten Style Option.
- PDF-Datei wird nicht direkt erstellt (PS-Datei muss nachträglich umgewandelt werden)
- **Gedrehte Texte in Grafik nur eingeschränkt möglich**

5 ODS PDF

Erstellt die PDF-Datei mit Hilfe des SAS PDF Treibers im „Universal Printing mode“ (identisch mit ODS PRINTER PDF), produktiv ab SAS Version 8.2.

Allgemeine Syntax:

```
ODS PDF
FILE=' '
STYLE=PRINTER          (Standard)
AUTHOR="text"
KEYWORDS="text"
SUBJECT="text"
TITLE="text"
STARTPAGE=NOW|NO|NEVER (Standard ist NOW)
; (NOTOC verhindert Table of Contents)

ODS PROCLABEL="text"; * Setzt TOC Label
```


Die Optionen **STYLE**, **AUTHOR**, **KEYWORDS**, **SUBJECT** füllen die entsprechenden Eigenschaftsfelder der PDF-Datei.

STARTPAGE legt fest, wann Seitenumbrüche erfolgen sollen.

NOW = mindestens nach jedem Outputbeitrag oder Grafik
 NO = lässt mehrere Textoutputs pro Seite zu
 NEVER = lässt Text- und Grafikoutputs pro Seite zu

Standardmäßig wird ein „Table of Contents“ (Bookmark) erstellt (abschaltbar mit NOTOC Option). Die dabei verwendeten Labels der Haupteinträge (normalerweise der Prozedurname) lassen sich mit ODS PROCLABEL, die der Untereinträge bei Grafikprozeduren über die DESCRIPTION= Option ändern.

Vorgehensweise:

1. PDF-Datei in ODS PDF Anweisung festlegen. Optional kann über die Option STYLE= das Layout beeinflusst werden. Standardmäßig wird STYLE=PRINTER verwendet.

```
ODS PDF FILE='C:\temp\test3.pdf';
```
2. Der Listen- und Grafikoutput aller nachfolgend aufgerufenen Prozeduren wird in die angegebene PDF-Datei geschrieben, bis diese explizit geschlossen wird. Wichtig ist es, Prozeduren mit RUN bzw. QUIT vor dem ODS CLOSE zu beenden, sonst fehlt der Output der letzten Prozedur in der PDF-Datei.

```
ODS PDF CLOSE;
```
3. Standardmäßig wird die Datei anschließend im SAS Results Viewer Fenster angezeigt.

Hinweise:

1. Standardmäßig wird bei ODS PDF das Letter Format verwendet, was in der PDF-Datei zu großen Seitenrändern und damit zu einer Verkleinerung des bedruckbaren Bereichs führt. Mit Hilfe der Systemoption **PAPERSIZE= "ISO A4"** (muss vor der ODS PDF Anweisung gesetzt sein) kann die PDF-Datei im A4 Format erstellt werden.
2. In Version 8.2 werden **gedrehte Texte** mit True Type Fonts (z.B. A=90) falsch dargestellt. Das Problem soll in Version 9 behoben sein. Siehe SAS Note SN-005255. Stattdessen können SAS/GRAPH Software Fonts wie SWISS verwendet werden.
3. Probleme mit URLs siehe
www.sas.com/rnd/base/topics/odsprinter/faq.html#badpdf82

Ergebnis:

- PDF-Datei wird direkt gestellt.
- Alle Outputs befinden sich in einer PDF-Datei, deren Größe hängt auch von der verwendeten Style Option ab (im Beispiel 162 KB).
- Jeder Eintrag und jede Seite wird auch als Lesezeichen (Table of Contents, Bookmark) eingefügt (abschaltbar mit NOTOC Option).
- Listenoutput wird hochaufgelöst als Tabelle erstellt, abhängig von der verwendeten Style Option.
- **Gedrehte Texte in Grafik nicht möglich.**

6 Zusammenfassung und Bewertung

Methode: Distiller ohne ODS	
Vorteil	Nachteil
PDF-Datei wird direkt erstellt.	Listen- und Grafikoutput in verschiedenen PDF-Dateien.
Listenoutput platzsparend.	Für jede Grafik muss einzeln eine PDF-Datei erstellt werden.
SWISS, SIMPLEX, etc. in Grafik verwendbar.	Grafik wird in PDF-Datei im Hochformat angezeigt.
Methode: ODS PRINTER PRINTER='Acrobat Distiller'	
Vorteil	Nachteil
Listen- und Grafikoutput in einer PDF-Datei.	PDF-Datei wird nicht direkt erstellt.
Listenoutput hochaufgelöst (besseres Layout).	Listenoutput hochaufgelöst (großer Platzbedarf).
SWISS, SIMPLEX, etc. in Grafik verwendbar.	SWISS, SIMPLEX, etc. führt zu großen PDF-Dateien. Probleme mit gedrehten Texten und True Type Fonts in Grafik.
Methode: ODS PS	
Vorteil	Nachteil
Listen- und Grafikoutput in einer PDF-Datei.	PDF-Datei wird nicht direkt erstellt.
Listenoutput hochaufgelöst (besseres Layout).	Listenoutput hochaufgelöst (großer Platzbedarf).
True Type Fonts.	Gedrehte Texte mit True Type Fonts in Grafik nur eingeschränkt möglich.

Methode: ODS PDF	
Vorteil	Nachteil
Listenoutput hochaufgelöst (besseres Layout).	Listenoutput hochaufgelöst (großer Platzbedarf).
Listen- und Grafikoutput in einer PDF-Datei.	Keine SAS/GRAPH Software Fonts wie SWISS, SIMPLEX.
True Type Fonts.	Gedrehte Texte in Grafik fehlerhaft.
PDF-Datei wird direkt erstellt.	
Bookmarks (Table of Contents) pro Eintrag +Seite (abschaltbar mit NOTOC Option).	

7 Testprogramm

```

***** 1. Distiller (Windows, ohne ODS)*****;
OPTIONS SYSPRINT='Acrobat Distiller';
OPTIONS SYSPRINTFONT='SAS Monospace' 7;
OPTIONS ORIENTATION=PORTRAIT PAPERSIZE="ISO A4"
        LEFTMARGIN="1.0 CM" RIGHTMARGIN="0.5 CM"
        TOPMARGIN="0.5 CM" BOTTOMMARGIN="0.5 CM";
OPTIONS DATE NOCENTER;
GOPTIONS RESET=ALL DEV=WIN TARGET=WINPRTC ROTATE=LANDSCAPE
        GUNIT=PCT CTEXT=BLUE FTEXT=SWISS HTEXT=2.0 CBY=BLACK
        FBY=SWISS HBY=2.5 BORDER KEYMAP=WINANSI DEVMAP=WINANSI;

TITLE1 H=4.0 "Normaler PDF Output mit Distiller (no ODS)";
TITLE2 H=3.0 "SAS &SYSVLONG &SYSSCPL, Umlaute öäü ÖÄÜ ß";
FOOTNOTE1 F=SIMPLEX A=90 H=2.0 PCT C=BLACK "Umlaute öäü ÖÄÜ
        ß: &SYSDATE %SYSFUNC(TIME(),TIME5.)";

PROC SORT DATA=sashelp.column;
        BY type;
RUN;

PROC PRINT DATA=sashelp.column;
        VAR colname--defcred;
PROC GCHART DATA=sashelp.column;
        HBAR label; BY type;
RUN; QUIT;

```

```
***** 4. ODS PDF *****;
OPTIONS ORIENTATION=LANDSCAPE;
GOPTIONS ROTATE=LANDSCAPE GUNIT=PT FTEXT="Helvetica"
  FBY="Helvetica" HTEXT=12 PT HBY=12 PT;
TITLE1 F="Helvetica" H=14 PT C=BLUE "ODS PDF";
TITLE2 F="Times" H=12 PT C=BLUE "SAS &SYSVLONG
  &SYSSCPL, Umlaute öäü ÖÄÜ ß";
FOOTNOTE1 F="Helvetica" A=00 H=7 PT C=BLACK "Umlaute öäü
  ÖÄÜ ß: &SYSDATE %SYSFUNC(TIME()),TIME5.)";

ODS PRINTER PDF FILE='C:\temp\test3.pdf' TITLE="ODS PDF"
  SUBJECT="ODS_PDF.sas" AUTHOR="HS" STARTPAGE=NOW;
ODS LISTING CLOSE;

ODS PROCLABEL="Listing von PROC PRINT";
PROC PRINT DATA=sashelp.column;
  VAR colname--defcred;
RUN;
ODS PROCLABEL="Listing von PROC GCHART";
PROC GCHART DATA=sashelp.column;
  HBAR label / DES="Überschrift"; BY type;
RUN; QUIT;

ODS PRINTER CLOSE;
ODS LISTING;
```